


- 6.2 Try to walkthrough the following program and write down the expected results. Key-in the program and compare the results after execution. This program helps you to understand the function of exchanging the values of two variables and the concepts of pass-by-value and pass-by-reference.

**Step 1:** Create a form with two command buttons and one picture box according to the properties table below

### Properties Table

Object	Property	Setting
Form	Name	frmExchange
	Caption	Number Exchange
Command Button	Name	cmdExchange
	Caption	&Exchange Numbers
Command Button	Name	cmdExit
	Caption	&Exit
Picture Box	Name	picOutput
	Caption	(empty)

### Layout



**Step 2 :**     *Add codes for the events*  
                   *cmdExchange\_Click()*  
                   *cmdExit*

Codes for **cmdExchange**

```
Private Sub cmdExchange_Click()
    Dim A As Integer
    Dim B As Integer
    Dim Temp As Integer

    picOutput.Cls
    picOutput.Print "Start program."
    picOutput.Print

    A = 1
    B = 2
    Temp = 3

    picOutput.Print Tab(10); "A"; Tab(20); "B"; Tab(30); "Temp"
    picOutput.Print Tab(10); A; Tab(20); B; Tab(30); Temp
    Call Exchange1(A, B)
    picOutput.Print Tab(10); A; Tab(20); B; Tab(30); Temp
    Call Exchange2(A, B)
    picOutput.Print Tab(10); A; Tab(20); B; Tab(30); Temp
    Call Exchange2(A, B)
    picOutput.Print Tab(10); A; Tab(20); B; Tab(30); Temp
    picOutput.Print
    picOutput.Print "End program."
End Sub
```

Codes for **cmdExit**

```
Private Sub cmdExit_Click()
    End
End Sub
```

**Step 3 :**     *Add codes for the subprograms* **Exchange1** *and* **Exchange2**

```
Private Sub Exchange1(ByVal A As Integer, ByVal B As Integer)
    Dim Temp As Integer
    Temp = A
    A = 2 * B
    B = 2 * Temp

End Sub

Private Sub Exchange2(A As Integer, B As Integer)
    Dim Temp As Integer
    Temp = A
    A = 2 * B
    B = 2 * Temp
    picOutput.Print Tab(10); A; Tab(20); B; Tab(30); Temp

End Sub
```

**Step 4 :**     *Execution*

Click the **Exchange Numbers** command button.  
 Terminate the program by clicking the **Exit** button